A few Mathematica gotchas

Put spaces between factors in products

To write an expression like $v\,t$, be sure to put a space between v and t. Without the space, "vt" is interpreted as a single symbol.

```
In[1]:= D[v t, t]
```

$$v$$

```
In[2]:= D[vt, t]
```

$$0$$

You should be able to see the difference in font: Mathematica will render $v\,t$ with the space but vt without it. Here are a few more examples:

```
In[3]:= {mc^2, m c^2} / m
```

$$\left\{\frac{mc^2}{m},\ c^2\right\}$$

```
In[4]:= {Integrate[Cos[at], t], Integrate[Cos[a t], t]}
```

$$\left\{t\cos(at),\ \frac{\sin(a\,t)}{a}\right\}$$

Use Exp[x] or E^x for the exponential function e^x

- "e" *vs* "E"

The symbol lower case "e" has no special meaning in Mathematica. The capital letter "E" means Euler's constant $e = 2.71828\ldots$.

Since the symbol "e" is a variable with no value assigned, Mathematica can't find it's numerical value.

```
In[5]:= {e, N[e]}
```

$$\{e,\ e\}$$

"E" is Mathematica's built-in symbol for Euler's constant

```
In[6]:= {E, N[E]}
```

$$\{e,\ 2.71828\,18284\,5905\}$$

Notice the *slight* difference in font in the output: the variable e vs the mathematical constant e.

- ### E^x *vs* e^x

Since “E” is Euler’s constant, “E^x” is the exponential function.

This expression computes the derivative of the exponential function.

```
In[7]:= D[E^x, x]
Out[7]= e^x
```

This next expression computes the derivative of the variable “e” to the power *x*. Since “e” has no special meaning, Mathematica doesn’t know that you meant it to be Euler’s constant and doesn’t recognize this as the exponential function.

```
In[8]:= D[e^x, x]
Out[8]= e^x log(e)
```

I usually write Exp[x] instead of E^x, but either will work.

```
In[9]:= D[Exp[x], x]
Out[9]= e^x
```

Avoid built-in constants

The capital letters C, D, E, I, N, and O have built-in meanings and can’t be used for variable names. This is annoying because those letters are used in conventional notation for many problems: *E* for the energy or electric field, *I* for the identity matrix in linear algebra or the current in electrical engineering, and so on.

The obvious symbol for energy won’t work:

```
In[10]:= E = m c^2
... Set: Symbol e is Protected.
Out[10]= c^2 m
```

The assignment couldn’t be done, so “E” remains set to Euler’s constant, not $m\,c^2$.

```
In[11]:= E
Out[11]= e
```

Likewise for *I*:

```
In[12]:= I = {{1, 0}, {0, 1}}
... Set: Symbol i is Protected.
Out[12]= {{1, 0}, {0, 1}}
```

This is one of my few serious complaints about *Mathematica.* Hardwiring meanings to commonly-used letters isn’t cool; they could have used “EulerE” instead of “E”, “Diff” instead of “D”, and so on.

As a workaround I usually use “EE”, “DD”, and so on when I want variables “E” and “D”.

• A “bright idea” with awful side effects.

Mathematica will let you overwrite protected symbols. However, it’s a terrible idea to do this.

Let’s remove the built-in definition of “E”, letting me use it for energy.

```
In[13]:= Unprotect[E]
Out[13]= {E}
```

Woo-hoo, now I can use “E” for energy!

```
In[14]:= E = m c^2
Out[14]= c^2 m
```

But there’s a side effect:

```
In[15]:= Exp[x]
Out[15]= (c^2 m)^x
```

That’s probably not what you wanted. Running “E=.” will restore the old definition.

```
In[16]:= E = .

In[17]:= E
Out[17]= e
```

Let’s get back to safety, protecting Euler’s constant

```
In[18]:= Protect[E]
Out[18]= {E}

In[19]:= Exp[x]
Out[19]= e^x
```