Substitution rules

Katharine Long, Texas Tech University

In our course we'll use substitution rules primarily for minor simplifications and for plugging solutions or numerical values into expressions. There's much more that can be done with them; if you want to find out more, see `http://reference.wolfram.com/language/guide/RulesAndPatterns.html`.

Basics of substitution rules

As the name implies, substitution rules are used to replace one expression with another. Here's how to replace x with y^2 in the expression $1+x$:

In[∘]:= `1 + x /. x → y^2`

Out[∘]= $y^2 + 1$

The "/." operator (pronounced "slash-dot") applies the rule $x \to y^2$ to the expression $1 + x$. To enter the arrow "→", you can either type a dash "-" and then a greater-than ">", or type ESC->ESC. I usually do the former.

You can replace any expression with any other expression. For example,

In[∘]:= `1 + x /. x → -Cos[θ]^2`

Out[∘]= $1 - \cos^2(\theta)$

In[∘]:= `Exp[x] /. x → Integrate[f[t], t]`

Out[∘]= $e^{\int f(t)\,dt}$

You can replace a more complicated expression by a less complicated expression

In[∘]:= `1 / Sqrt[1 + Exp[x]] /. Exp[x] → u`

Out[∘]= $\frac{1}{\sqrt{u+1}}$

• A practical example: plugging numbers into a formula

You'll usually want to go as far as possible in a calculation before plugging in numbers for symbols. Here's a calculation of the velocity "kick" Δv given to an alpha particle during a close encounter with a gold atom in the Rutherford experiment. The distance of closest approach ("impact parameter") is *b*, and the initial velocity is *v*. The calculation is done using the "impulse approximation" (an exact calculation can be done using methods you'll learn in classical mechanics.)

```
In[∘]:= Δv = Z₁ Z₂ e^2 / (4 Pi ϵ₀ mα) Integrate[b / (b^2 + v^2 t^2)^(3/2),
           {t, -Infinity, Infinity}, Assumptions → {v > 0, b > 0}]
```

Out[∘]= $$\frac{e^2 Z_1 Z_2}{2 \pi b v \epsilon_0 m_\alpha}$$

It's a really bad idea to plug in numbers *before* doing the integral. However, to compare to data we need to plug numbers into the result, and you'll often need to play around with parameters. That's easily done with substitution rules.

At the time of Rutherford's experiment, the atom was imagined to be a distributed blob of positive charge with electron particles within ("plum pudding model"). The radius of a gold atom was thought to be about 10^{-10}m, so the maximum force would occur at about $b \approx 10^{-10}$m. The deflection angle is $\Delta\theta \approx \tan^{-1}(\Delta v / v)$. Plugging in numbers and converting to degrees, we find:

```
In[∘]:= Δθ = 180 / Pi ArcTan[Δv / v] /.  {b → 10^-10, e → 1.602 × 10^-19,
          Z₁ → 79, Z₂ → 2, v → 1.5 × 10^7, ϵ₀ → 8.854 × 10^-12, mα → 6.645 × 10^-27}
```

Out[∘]= 0.0279323721140597

That's under a degree; Rutherford and his students Geiger and Marsden found that some particles were being scattered by over 90°.

Try again with a smaller b, say $b = 10^{-14}$m:

```
In[∘]:= Δθ = 180 / Pi ArcTan[Δv / v] /.  {b → 10^-14, e → 1.602 × 10^-19,
          Z₁ → 79, Z₂ → 2, v → 1.5 × 10^7, ϵ₀ → 8.854 × 10^-12, mα → 6.645 × 10^-27}
```

Out[∘]= 78.40809 98899 838

That's much closer to the observed large angle scattering. In your advanced courses you'll do a more accurate calculation.

• A practical example: simplifying Fourier coefficients

In Fourier analysis, we'll need to do integrals like this one.

```
In[∘]:= Bn_ = 1 / Pi Integrate[x Sin[n x], {x, -Pi, Pi}]
```

Out[∘]= $$\frac{2 \sin(\pi n) - 2 \pi n \cos(\pi n)}{\pi n^2}$$

In that application, the variable n is a positive integer; therefore we know that $\sin(n\,\pi) = 0$ and $\cos(n\,\pi) = (-1)^n$. But Mathematica doesn't know that restriction on n, so produced the general result.

To clean up the formula, use a substitution rule to replace $\sin(n\,\pi)$ by 0 and $\cos(n\,\pi)$ by $(-1)^n$.

```
In[∘]:= Bn /. {Sin[n Pi] → 0, Cos[n Pi] → (-1)^n}
```

Out[∘]= $$-\frac{2 (-1)^n}{n}$$

That's much more readable.