Working with units

Katharine Long, Texas Tech University

Units are helpful for catching mistakes.

Example: free fall time

The free fall time of an object dropped from height *h* is:

In[1]:=
```
Tfall = Sqrt[2 h / g]
```

Out[1]= $\sqrt{2}\sqrt{\frac{h}{g}}$

We can plug in numbers using substitution rules,

In[2]:=
```
Tfall /. {h → 10, g → 9.81}
```

Out[2]= 1.42784 31229 2706

However, it’s helpful to work with units. That’s done using the Quantity function. The units are given by strings: “Meter”, “Second”, “Kilogram”, etc.

In[3]:=
```
Tfall /. {h → Quantity[10, "Meter"], g → Quantity[9.81, "Meter/Second/Second"]}
```

Out[3]= 1.42784 31229 2706 s

Mathematica is flexible about how you specify units.

In[12]:=
```
Tfall /. {h → Quantity[10, "mtr"], g → Quantity[9.81, "m /sec^2"]}
```

Out[12]= 1.42784 31229 2706 s

• Catching mistakes

Including units is helpful for catching mistakes. Suppose I goofed and wrote $T = \sqrt{2\,h/m\,g}$ instead of $T = \sqrt{2\,h/g}$.

In[14]:=
```
Tdoh = Sqrt[2 h / g / m]
```

Out[14]= $\sqrt{2}\sqrt{\frac{h}{g\,m}}$

In[18]:=
```
Tdoh /. {h → 10, g → 9.81, m → 0.1}
```

Out[18]= 4.51523 64098 5731

It’s hard to spot the error in a pure number.

In[17]:=
```
T_doh /. {h → Quantity[10, "Meter"],
  g → Quantity[9.81, "Meter/Second/Second"], m → Quantity[0.1, "kg"]}
```

Out[17]= $4.51523\,64098\,5731\ \mathrm{s}/\sqrt{\mathrm{kg}}$

The units came out wrong: a big flashing red light telling me I’ve made a mistake.

Example: Rutherford scattering

In the SubstitutionRules document, we worked through an estimate of the scattering angle in Rutherford’s gold foil experiment.

In[5]:=
```
Δv = Z1 Z2 e^2 / (4 Pi ϵ0 mα) Integrate[b / (b^2 + v^2 t^2) ^ (3 / 2),
   {t, -Infinity, Infinity}, Assumptions → {v > 0, b > 0}]
```

Out[5]= $$\frac{e^2 Z_1 Z_2}{2\,\pi\,b\,v\,\epsilon_0\,m_\alpha}$$

We plugged in numbers for the “plum pudding” model

In[6]:=
```
Δθ = 180 / Pi ArcTan[Δv / v] /. {b → 10^-10, e → 1.602 × 10^-19,
   Z1 → 79, Z2 → 2, v → 1.5 × 10^7, ϵ0 → 8.854 × 10^-12, mα → 6.645 × 10^-27}
```

Out[6]= 0.02793 23721 140597

It’s easy to make a mistake in typing in all those numbers, and it’s hard to catch any mistakes we might have made in deriving the formula. Units to the rescue.

In[19]:=
```
Δθ = Quantity[180 / Pi, "Degree"] ArcTan[Δv / v] /. {
   b → Quantity[10^-10, "Meter"],
   e → Quantity[1.602 × 10^-19, "Coulomb"], Z1 → 79, Z2 → 2,
   v → Quantity[1.5 × 10^7, "Meter/Second"],
   ϵ0 → Quantity[8.854 × 10^-12, "Farad/Meter"],
   mα → Quantity[6.645 × 10^-27, "Kilogram"]}
```

Out[19]= 0.02793 23721 140597°

The result is in degrees, as expected.

Following Rutherford, try again with a smaller b, say $b = 10^{-14}$m:

In[21]:=
```
Δθ = Quantity[180 / Pi, "Degree"] ArcTan[Δv / v] /. {
   b → Quantity[10^-14, "Meter"],
   e → Quantity[1.602 × 10^-19, "Coulomb"], Z1 → 79, Z2 → 2,
   v → Quantity[1.5 × 10^7, "Meter/Second"],
   ϵ0 → Quantity[8.854 × 10^-12, "Farad/Meter"],
   mα → Quantity[6.645 × 10^-27, "Kilogram"]}
```

Out[21]= 78.40809 98899 838°

Example: electromagnetic skin depth

An electromagnetic wave cannot penetrate far into a conductor: the varying EM fields cause eddy currents which lose energy because of finite conductivity in the material. In an EM course you'll derive the "skin depth", which at low frequencies is $\delta = \sqrt{\frac{2}{\sigma\omega\mu}}$; here σ is the conductivity, ω is the angular frequency of the EM wave, and μ is the magnetic permeability of the material. Engineers often work with cyclical frequency f rather than angular frequency $\omega = 2\pi f$, so I'll write the expression in terms of f

In[25]:=
```
δ = Sqrt[2 / (σ 2 π f μ)]
```

Out[25]= $\dfrac{\sqrt{\frac{1}{f\mu\sigma}}}{\sqrt{\pi}}$

Skin depth for copper at 1 KHz:

In[42]:=
```
δCu =
 δ /. {σ → Quantity[5.96 × 10 ^7, "Siemen/Meter"], f → Quantity[1000, "Hertz"],
   μ → Quantity[4 Pi 10^-7, "Henry/Meter"]}
```

Out[42]= $0.00206\,15648\,54528\,45\ \mathrm{m}/(\sqrt{\mathrm{Hz}}\ \sqrt{\mathrm{H}}\ \sqrt{\mathrm{S}})$

Mathematica needs a little more help to convert this to base units.

In[43]:=
```
UnitSimplify[δCu]
```

Out[43]= $0.00206\,15648\,54528\,45\ \mathrm{m}$

The skin depth is about 2 mm. If you'd like to see the results explicitly in millimeters, that's easily done with the UnitConvert function

In[44]:=
```
UnitConvert[δCu, "Millimeter"]
```

Out[44]= $2.06156\,48545\,2845\ \mathrm{mm}$

You can convert to any crazy length unit you like

In[48]:=
```
{UnitConvert[δCu, "Inch"],
 UnitConvert[δCu, "Light Year"], UnitConvert[δCu, "Angstrom"]}
```

Out[48]= $\{0.08116\,39706\,50726\,5\ \mathrm{in},\ 2.17907\,57706\,3241\times 10^{-19}\ \mathrm{ly},\ 2.06156\,48545\,2845\times 10^{7}\ \text{Å}\}$

You'll be unable to convert to a unit that isn't a length:

In[46]:=
```
UnitConvert[δCu, "Week"]
```

Quantity: $\dfrac{\text{Meters}}{\sqrt{\text{Henries}}\ \sqrt{\text{Hertz}}\ \sqrt{\text{Siemens}}}$ and Weeks are incompatible units

Out[46]= $Failed

Plotting with units

The Plot function is aware of units.

In[30]:=
```
skinDepth[ν_] =
 UnitSimplify[δ /. {σ → Quantity[5.96 × 10 ^7, "Siemen/Meter"], f → ν,
    μ → Quantity[4 Pi 10^-7, "Henry/Meter"]}]
```

Out[30]=
$$\frac{\sqrt{\frac{0.01335\,19247\,56031\,5\ \mathrm{m^2/s}}{\nu}}}{\sqrt{\pi}}$$

In[36]:=
```
Plot[skinDepth[ν],
 {ν, Quantity[10, "Hertz"], Quantity[10 000, "Hertz"]}, FrameLabel → Automatic]
```

Out[36]=


Plot will catch incompatible units

In[37]:=
```
Plot[skinDepth[ν],
 {ν, Quantity[10, "Hertz"], Quantity[10 000, "Acres"]}, FrameLabel → Automatic]
```

••• Plot: Endpoints for ν in {ν, 10 Hz, 10000 acres} must have distinct machine−precision numerical values.

••• Quantity: Acres and Hertz are incompatible units

••• Plot: Limiting value QuantityMagnitude[$Failed] in {ν, 10, QuantityMagnitude[$Failed]} is not a machine−sized real number.

Out[37]= Plot[skinDepth(ν), {ν, 10 Hz, 10 000 acres}, FrameLabel → Automatic]