Underscores for arguments

Katharine Long, Texas Tech University

Here's the TL;DR summary:

Rule of thumb #1: you'll usually use underscores after arguments when defining functions unless you have a good reason not to.

Rule of thumb #2: ***never*** *put underscores after anything on the RHS of an assignment!*

My short explanations and examples are below. The gory details can be found at
`http://reference.wolfram.com/language/guide/Patterns.html`

- ## With an underscore: argument x is a variable whose value can be replaced by any expression

```
In[2]:= f[x_] = Exp[-x] Sin[x]
```

Out[2]= $e^{-x} \sin(x)$

This behaves like you think a function should: invoke $f(y)$, and it applies the function f to the argument y.

```
In[27]:= {f[2], f[y], f[x], f[x^2], f[u[x]]}
```

Out[27]= $\left\{\frac{\sin(2)}{e^2}, e^{-y} \sin(y), e^{-x} \sin(x), e^{-x^2} \sin\left(x^2\right), e^{-u(x)} \sin(u(x))\right\}$

- ## Without an underscore: define the function for that specific symbolic argument *only*

```
In[29]:= g[x] = Exp[-x] Sin[x]
```

Out[29]= $e^{-x} \sin(x)$

This defines g only when the argument is exactly the symbol x.

```
In[28]:= {g[2], g[y], g[x], g[x^2], g[u[x]]}
```

Out[28]= $\left\{g(2), g(y), e^{-x} \sin(x), g\left(x^2\right), g(u(x))\right\}$

This probably wasn't what you wanted.

Why would you ever *not* use an underscore?

You omit the underscore only when you want to set values of a function for certain special arguments.

Example: define the factorial recursively

The factorial is defined by $0! = 1$ and then $n! = n(n-1)!$ for $n > 0$.

In[16]:=
```
Remove[fact]
```

Write the function for generic *n*. Use an underscore here. Notice the use of deferred assignment (":="), so that we don't get thrown into infinite recursion here.

In[30]:=
```
fact[n_] := n fact[n - 1]
```

Set the special value at n=0. No underscore. We can use immediate assignment.

In[18]:=
```
fact[0] = 1
```

Out[18]= 1

We're done! Compute some factorials.

In[21]:=
```
Table[fact[n], {n, 0, 10}]
```

Out[21]= {1, 1, 2, 6, 24, 120, 720, 5040, 40 320, 362 880, 3 628 800}

Example: Construction of the Legendre polynomials

The Legendre polynomials are defined by a recurrence relation

$$(n+1)\,P_{n+1}(x) = (2\,n+1)\,x\,P_n(x) - n\,P_{n-1}(x)$$

with initial values $P_0(x) = 1$ and $P_1(x) = x$. They're useful in physics problems set in spherical coordinates, and also appear in computational methods.

We want to write a function p(n,x) that will evaluate $P_n(x)$ for the index *n* and the variable *x*.

Set the initial values for $n = 0$ and $n = 1$. Since *x* will be the variable, we still use underscores for *x* even though we're setting values for special values of *n*.

```
p[0, x_] = 1; p[1, x_] = x;
```

Write the recurrence relation. We use underscores for both arguments, since this is the definition to be used for "generic" values of *n*. As before, we use underscores for *x*.

In[23]:=
```
p[n_, x_] := ((2 n - 1) x p[n - 1, x] - (n - 1) p[n - 2, x]) / n
```

In[∘]:=
```
p[2, x]
```

Out[∘]= $\frac{1}{2}\left(3\,x^2 - 1\right)$

In[∘]:=
```
p[3, x] // FullSimplify
```

Out[∘]= $\frac{1}{2}\,x\left(5\,x^2 - 3\right)$

In[25]:=
```
Table[p[n, x], {n, 1, 4}]
```

Out[25]=
$$\left\{x,\ \frac{1}{2}\left(3x^2-1\right),\ \frac{1}{3}\left(\frac{5}{2}x\left(3x^2-1\right)-2x\right),\ \frac{1}{4}\left(\frac{7}{3}x\left(\frac{5}{2}x\left(3x^2-1\right)-2x\right)-\frac{3}{2}\left(3x^2-1\right)\right)\right\}$$

In[26]:=
```
Table[p[n, x], {n, 1, 4}] // FullSimplify
```

Out[26]=
$$\left\{x,\ \frac{1}{2}\left(3x^2-1\right),\ \frac{1}{2}x\left(5x^2-3\right),\ \frac{1}{8}\left(35x^4-30x^2+3\right)\right\}$$

- Prettier notation for the Legendre polynomials

In[34]:=
```
Remove[P]
```

In[35]:= P_0[x_] = 1; P_1[x_] = x;

In[36]:= $P_{n\_}$[x_] := ((2 n - 1) x P_{n-1}[x] - (n - 1) P_{n-2}[x]) / n

In[37]:= Table[P_n[x], {n, 0, 4}] // FullSimplify

Out[37]=
$$\left\{1,\ x,\ \frac{1}{2}\left(3x^2-1\right),\ \frac{1}{2}x\left(5x^2-3\right),\ \frac{1}{8}\left(35x^4-30x^2+3\right)\right\}$$